

## 4.3 ML<sup>3</sup>の型推論

### 4.3.1 関数に関する型付け規則

次に, fun 式, 関数適用式

$$e ::= \dots \mid \text{fun } x \rightarrow e \mid e_1 e_2$$

で型推論アルゴリズムを拡張しよう。「 $\tau_1$  の値を受け取って (計算が停止すれば)  $\tau_2$  の値を返す関数」の型を  $\tau_1 \rightarrow \tau_2$  とすると, 型の定義は以下のように変更される.

$$\tau ::= \text{int} \mid \text{bool} \mid \tau_1 \rightarrow \tau_2$$

これらの式に関して型付け規則は以下のように与えられる.

$$\frac{\Gamma, x : \tau_1 \vdash e : \tau_2}{\Gamma \vdash \text{fun } x \rightarrow e : \tau_1 \rightarrow \tau_2} \quad (\text{T-FUN})$$

$$\frac{\Gamma \vdash e_1 : \tau_1 \rightarrow \tau_2 \quad \Gamma \vdash e_2 : \tau_1}{\Gamma \vdash e_1 e_2 : \tau_2} \quad (\text{T-APP})$$

それぞれの規則について簡単に説明を加える.

**T-FUN:** 引数  $x$  が  $\tau_1$  を持つという仮定の下で関数本体  $e$  が  $\tau_2$  型を持つならば,  $\text{fun } x \rightarrow e$  が  $\tau_1 \rightarrow \tau_2$  型を持つことを導いて良い. これは関数のセマンティクスから理解できるであろう.

**T-APP:**  $e_1$  の型が関数型  $\tau_1 \rightarrow \tau_2$  であり, かつ, その引数の型  $\tau_1$  と  $e_2$  の型が一致している場合に, 適用式全体に  $e_1$  の返り値型  $\tau_2$  がつくことを導いて良い. これも関数型の直観と関数適用のセマンティクスから分かるであろう.

次は型推論アルゴリズムの設計である. これらの規則を含めても型付け規則は構文主導なので, 前節の「規則を下から上に読む」という戦略を使ってみよう. 入力として型環境  $\Gamma$  と式  $e$  が与えられ, 式  $e$  が  $\text{fun } x \rightarrow e_1$  という形をしていたとしよう. そうすると, T-FUN を下から上に使うことに読んで, 以下のように型推論ができそうである.

1. 型環境  $\Gamma, x : \tau_1$  と式  $e_1$  を入力として型推論アルゴリズムを再帰的に呼び出し型  $\tau_2$  を得る.
2. 型  $\tau_1 \rightarrow \tau_2$  を  $e$  の型として返す.

ところが, これでは型推論が実装できない. 問題は, 最初のステップで  $e_1$  の型を調べる際に作る型環境  $\Gamma, x : \tau_1$  である. ここで  $x$  の型として  $\tau_1$  を取っているが, この型をどのように

取るべきかは、一般には  $e_1$  の中での  $x$  の使われ方と、この関数  $\text{fun } x \rightarrow e_1$  がどのように使われうるかに依存するので、このタイミングで  $\tau_1$  を容易に決めることはできない。

簡単な例として、 $\text{fun } x \rightarrow x + 1$  という式を考えてみよう。これは、 $\text{int} \rightarrow \text{int}$  型の関数であることは「一目で」わかるので、一見、 $x$  の型を  $\text{int}$  として推論を続ければよさそうだが、問題は、本体式である  $x + 1$  を見るまえには、 $x$  の型が  $\text{int}$  であることがわからない。

### 4.3.2 型変数、型代入と型推論アルゴリズムの仕様

この「 $\tau_1$  の適切な取り方が後にならないとわからない」という問題を解決するために「今のところ正体がわからない未知の型」を表す型変数 (*type variable*) を導入しよう。

$$\tau ::= \alpha \mid \text{int} \mid \text{bool} \mid \tau_1 \rightarrow \tau_2$$

そして、型推論アルゴリズムの出力として、入力（正確には型環境中）に現れる型変数の「正体が何か」を返すことにする。上の例だと、とりあえず  $x$  の型は  $\alpha$  などと置いて、型推論を続ける。推論の結果、 $x + 1$  の型は  $\text{int}$  である、という情報に加え  $\alpha = \text{int}$  という「型推論の結果  $\alpha$  は  $\text{int}$  であることが判明しました」という情報が返ってくることになる。最終的に T-FUN より、全体の型は  $\alpha \rightarrow \text{int}$ 、つまり、 $\text{int} \rightarrow \text{int}$  であることがわかる。また、 $\text{fun } x \rightarrow \text{fun } y \rightarrow x y$  のような式を考えると、以下のような手順で型推論がすすむ。

1. 新しい（つまり、他の型変数とカブらない）型変数  $\alpha$  を生成し、 $x$  の型を  $\alpha$  と置いて、本体、つまり  $\text{fun } y \rightarrow x y$  の型推論を行う。
2. 新しい型変数  $\beta$  を生成し、 $y$  の型を  $\beta$  と置いて、本体、つまり  $x y$  の型推論を行う。
3.  $x y$  の型推論の結果、この式の型が別の新しい型変数  $\gamma$  を使って  $\beta \rightarrow \gamma$  と書け、 $\alpha = \beta \rightarrow \gamma$  であることが判明する。

さらに詳しい型推論アルゴリズムの中身については後述するが、ここで大事なことは、とりあえず未知の型として用意した型変数の正体が、推論の過程で徐々に明らかになっていくことである。<sup>1</sup>

**型変数と多相型** OCaml の多相型 (*polymorphic type*) とここで導入する型変数を含む型とを混同してはならない。OCaml における多相型は、例えば OCaml では  $\text{fun } x \rightarrow x$  が型  $'a \rightarrow 'a$  を持ち、ここで表示される  $'a$  を「型変数」ということがあるが、この  $'a$  は上記の型変数とは異なる。この  $'a$  は「何の型にでも置き換えてよい」変数であるが、上記の型変数は「特定の型を表す」記号である。「任意の型で置き換え可能」な型変数は次節で再登場する。

<sup>1</sup>余談ではあるが、ここで用いられている方法は、未知の情報を含む問題を解くために (1) 未知の情報をとりあえず変数において (2) その変数が満たすべき制約を生成し (3) その制約を解くという、より一般的な問題解決の手法の適用例と見ることができる。方程式を立ててそれを解くとか、散々やってきたよね？

ここまで述べたことを実装したのが図 4.1 である．型 `ty` を型変数を表すコンストラクタ `TyVar` と関数型を表すコンストラクタ `TyFun` とで拡張する．`TyVar` は `tyvar` 型の値を一つとるコンストラクタで `TyVar(tv)` という形をしており，これが型変数を表す．`tyvar` 型は型変数の名前を表す型で，実体は整数型である．`TyFun` は `ty` 型の引数を 2 つ持つ `TyFun(t1,t2)` という形をしており，これが型  $\tau_1 \rightarrow \tau_2$  を表す．

型推論アルゴリズムの実行中には，他のどの型変数ともかぶらない新しい型変数を生成する必要がある．(このような型変数を *fresh* な型変数と呼ぶ．) これを行うのが関数 `fresh_tyvar` である．この関数は引数として `()` を渡すと (すなわち `fresh_tyvar ()` のように呼び出すと) 新しい未使用の型変数を生成する．この関数は次に生成すべき型変数の名前を表す整数への参照 `counter` を保持しており，保持している値を新しい型変数として返し，同時にその参照をインクリメントする．上で説明したように，`T-FUN` のケースでは新しい型変数を生成するのだが，その際にこの関数を使用する.<sup>2</sup>

上述の型変数とその正体の対応関係を，型代入 (*type substitution*) と呼ぶ．型代入 (メタ変数として  $S$  を使用する.) は，型変数から型への (定義域が有限集合な) 写像である．以下では， $S_\tau$  で  $\tau$  中の型変数を  $S$  を使って置き換えたような型， $S\Gamma$  で，型環境中の全ての型に  $S$  を適用したような型環境を表す．例えば  $S$  が  $\{\alpha \mapsto \text{int}, \beta \mapsto \text{bool}\}$  であるとき， $S\alpha = \text{int}$  であり， $S(\alpha \rightarrow \beta) = \text{int} \rightarrow \text{bool}$  であり， $S(x:\alpha, y:\beta) = (x:\text{int}, y:\text{bool})$  である． $S_\tau$ ,  $S\Gamma$  はより厳密には以下のように定義される．

$$\begin{aligned} S\alpha &= \begin{cases} S(\alpha) & \text{if } \alpha \in \text{dom}(S) \\ \alpha & \text{otherwise} \end{cases} \\ S\text{int} &= \text{int} \\ S\text{bool} &= \text{bool} \\ S(\tau_1 \rightarrow \tau_2) &= S\tau_1 \rightarrow S\tau_2 \\ \\ \text{dom}(S\Gamma) &= \text{dom}(\Gamma) \\ (S\Gamma)(x) &= S(\Gamma(x)) \end{aligned}$$

$S\alpha$  のケースが実質的な代入を行っているケースである． $S$  の定義域  $\text{dom}(S)$  に  $\alpha$  が入っている場合は， $S$  によって定められた型 (すなわち  $S\alpha$ ) に写像する．`int` と `bool` は型変数を含まないので， $S$  を適用しても型に変化はない． $\tau$  が  $\tau_1 \rightarrow \tau_2$  であった場合は再帰的に  $S$  を適用する．

型代入を使うと，新しい型推論アルゴリズムの仕様は以下のように与えられる．

入力: 型環境  $\Gamma$  と式  $e$

出力:  $S\Gamma \vdash e : \tau$  を結論とする判断が存在するような型  $\tau$  と代入  $S$

<sup>2</sup>関数 `fresh_tyvar` は呼び出すたびに異なる値を返すことに注意せよ．これは `fresh_tryvar` が純粋な意味での計算ではない (参照の値の更新や参照からの値の呼び出しといった) 副作用 (*side effect*) を持つためである．

型推論アルゴリズムを実装する前に、以降で使う補助関数を定義しておこう。

**Exercise 4.3.1** 図 4.1 中の `pp_ty`, `freevar_ty` を完成させよ. `freevar_ty` は、与えられた型中の型変数の集合を返す関数で、型は

```
val freevar_ty : ty -> tyvar MySet.t
```

とする. 型 `'a MySet.t` は `mySet.mli` で定義されている `'a` を要素とする集合を表す型である.

さて、型推論アルゴリズムを実装するためには、型代入を表すデータ構造を決める必要がある. 様々な表現方法がありうるが、ここでは素直に型変数と型のペアのリストで表現することにしよう. すなわち、型代入を表す OCaml の型は以下のように宣言された `subst` である.

```
type subst = (tyvar * ty) list
```

`subst` 型は `[(id1,ty1); ...; (idn,tyn)]` の形をしたリストである. このリストは `[idn ↦ tyn] ∘ ... ∘ [id1 ↦ ty1]` という型代入を表すものと約束する. つまり、この型代入は「受け取った型中の型変数 `id1` をすべて型 `ty1` に置き換え、得られた型中の型変数 `id2` をすべて型 `ty2` に置き換え... 得られた型中の型変数 `idn` をすべて型 `tyn` に置き換える」ような代入である. リスト中の型変数と型のペアの順序と、代入としての作用の順序が逆になっていることに注意してほしい. また、リスト中の型は後続のリストが表す型代入の影響を受けることに注意してほしい. 例えば、型代入 `[(alpha, TyInt)]` が型 `TyFun(TyVar alpha, TyBool)` に作用すると、`TyFun(TyVar TyInt, TyBool)` となり、型代入

```
[(beta, (TyFun (TyVar alpha, TyInt)))]; (alpha, TyBool)]
```

が型 `(TyVar beta)` に作用すると、まずリストの先頭の `(beta, (TyFun (TyVar alpha, TyInt)))` が作用して `TyFun (TyVar alpha, TyInt)` が得られ、次にこの型にリストの二番目の要素の `(alpha, TyBool)` が作用して `TyFun(TyBool, TyInt)` が得られる.

以下の演習問題で、型代入を作用させる補助関数を実装しよう.

**Exercise 4.3.2** 型代入に関する以下の型、関数を `typing.ml` 中に実装せよ.

```
type subst = (tyvar * ty) list
```

```
val subst_type : subst -> ty -> ty
```

例えば、

```
let alpha = fresh_tyvar () in
subst_type [(alpha, TyInt)] (TyFun (TyVar alpha, TyBool))
```

の値は `TyFun (TyInt, TyBool)` になり、

```
let alpha = fresh_tyvar () in
let beta = fresh_tyvar () in
subst_type [(beta, (TyFun (TyVar alpha, TyInt)))]; (alpha, TyBool)] (TyVar beta)
```

の値は `TyFun (TyBool, TyInt)` になる.

syntax.ml:

```
...
type tyvar = int

type ty =
  TyInt
  | TyBool
  | TyVar of tyvar
  | TyFun of ty * ty

(* pretty printing *)
let pp_ty = ...

let fresh_tyvar =
  let counter = ref 0 in
  let body () =
    let v = !counter in
    counter := v + 1; v
  in body

let rec freevar_ty ty = ... (* ty -> tyvar MySet.t *)
```

図 4.1:  $ML^3$  型推論の実装 (1)

### 4.3.3 単一化

型変数と型代入を導入したところで型付け規則をもう一度見てみよう。T-IF や T-PLUS などの規則は「条件式の型は `bool` でなくてはならない」「`then` 節と `else` 節の式の型は一致していなければならない」「引数の型は `int` でなくてはならない」という制約を課していることがわかる。

これらの制約を  $ML^2$  に対する型推論では、型 (すなわち `TyInt` などの定義される言語の型を表現した値) の比較を行うことでチェックしていた。例えば与えられた式  $e$  が  $e_1 + e_2$  の形をしていたときには、 $e_1$  の型  $\tau_1$  と  $e_2$  の型  $\tau_2$  を再帰的にアルゴリズムを呼び出すことにより推論し、それらが `int` であることをチェックしてから全体の型として `int` を返していた。

しかし、型の構文が型変数で拡張されたいま、この方法は不十分である。というのは、部分式の型 (上記の  $\tau_1$  と  $\tau_2$ ) に型変数が含まれるかもしれないからである。例えば、`fun x → 1 + x` という式の型推論過程を考えてみる。まず、 $\emptyset \vdash \text{fun } x \rightarrow 1 + x : \text{int} \rightarrow \text{int}$  であることに注意しよう。(実際に導出木を書いてチェックしてみることに。) したがって、型推論アルゴリズムは、この式の型として `int`  $\rightarrow$  `int` を返すように実装するのが望ましい。

では、空の型環境  $\emptyset$  と上記の式を入力として、型推論アルゴリズムがどのように動くべきかを考えてみよう。この場合、まず T-FUN を下から上に読んで、 $x$  の型を型変数  $\alpha$  とおいた型環境  $x : \alpha$  の下で  $1 + x$  の型推論をすることになる。その後、各部分式  $1$  と  $x$  の型を、アルゴリズムを再帰的に呼び出すことで推論し、`int` と  $\alpha$  を得る。 $ML^2$  の型推論では、ここでそれぞれの型が `int` であるかどうかを単純比較によってチェックし、`int` でなかったら型エラーを報告していた。しかし今回は後者の型が  $\alpha$  であって `int` ではないため、単純比較による部分式の型のチェックだけでは型推論が上手くいかない。

では、どうすれば良いのだろうか。定石として知られている手法は制約による型推論 (*constraint-based type inference*) という手法である。この手法では、与えられたプログラムの各部分式から型変数に関する制約 (*constraint*) が生成されるものと見て、式をスキャンする過程で制約を集め、その制約をあとで解き型代入を得る、という形で型推論アルゴリズムを設計する。例えば、上記の例では、「 $\alpha$  は実は `int` である」という制約が生成される。この制約を解くと型代入  $\{\alpha \mapsto \text{int}\}$  が得られる。

上記の場合は制約が単純だったが、T-APP で関数  $e_1$  の受け取る引数の型と  $e_2$  の型が一致すること、また T-IF で `then` 節と `else` 節の式の型が一致することを検査するためには、より一般的な、

与えられた型のペアの集合  $\{(\tau_{11}, \tau_{12}), \dots, (\tau_{n1}, \tau_{n2})\}$  に対して、 $S_{\tau_{11}} = S_{\tau_{12}}, \dots, S_{\tau_{n1}} = S_{\tau_{n2}}$  なる  $S$  を求めよ

という制約解消問題を解かなければいけない。このような問題は単一化 (*unification*) 問題と呼ばれ、型推論だけではなく、計算機による自動証明などにおける基本的な問題として知られている。例えば、 $\alpha$  と `int` は  $S(\alpha) = \text{int}$  なる型代入  $S$  により単一化できる。また、 $\alpha \rightarrow \text{bool}$  と  $(\text{int} \rightarrow \beta) \rightarrow \beta$  は  $S(\alpha) = \text{int} \rightarrow \text{bool}$  かつ  $S(\beta) = \text{bool}$  なる  $S$  により単一化できる。

単一化問題は、対象 (ここでは型) の構造や変数の動く範囲によっては、非常に難しくな

るが<sup>3</sup>, ここでは, 型が単純な木構造を持ち, 型代入も単に型変数に型を割当てただけのもの (一階の単一化 (*first-order unification*) と呼ばれる問題) なので, 解である型代入を求めるアルゴリズムが存在する.<sup>4</sup> (しかも, 求まる型代入がある意味で「最も良い」解であることがわかっている.)

一階の単一化を行うアルゴリズム  $\mathcal{U}(X)$  は, 型のペアの集合  $X$  を入力とし,  $X$  中のすべての型のペアを同じ型にするような型代入を返す. (そのような型代入が存在しないときにはエラーを返す.)  $\mathcal{U}$  は以下のように定義される.

$$\begin{aligned}
\mathcal{U}(\emptyset) &= \emptyset \\
\mathcal{U}(\{(\tau, \tau)\} \uplus X') &= \mathcal{U}(X') \\
\mathcal{U}(\{(\tau_{11} \rightarrow \tau_{12}, \tau_{21} \rightarrow \tau_{22})\} \uplus X') &= \mathcal{U}(\{(\tau_{11}, \tau_{21}), (\tau_{12}, \tau_{22})\} \uplus X') \\
\mathcal{U}(\{(\alpha, \tau)\} \uplus X') \quad (\text{if } \tau \neq \alpha) &= \begin{cases} \mathcal{U}([\alpha \mapsto \tau]X') \circ [\alpha \mapsto \tau] & (\alpha \notin FTV(\tau)) \\ \text{error} & (\text{その他}) \end{cases} \\
\mathcal{U}(\{(\tau, \alpha)\} \uplus X') \quad (\text{if } \tau \neq \alpha) &= \begin{cases} \mathcal{U}([\alpha \mapsto \tau]X') \circ [\alpha \mapsto \tau] & (\alpha \notin FTV(\tau)) \\ \text{error} & (\text{その他}) \end{cases} \\
\mathcal{U}(\{(\tau_1, \tau_2)\} \uplus X') &= \text{error} \quad (\text{その他の場合})
\end{aligned}$$

ここで,  $\emptyset$  は空の型代入を表し,  $[\alpha \mapsto \tau]$  は  $\alpha$  を  $\tau$  に写す (そしてそれ以外の型変数については何も行わない) 型代入である. また  $FTV(\tau)$  は  $\tau$  中に現れる型変数の集合である. また,  $X \uplus Y$  は,  $X \cap Y = \emptyset$  のときの  $X \cup Y$  を表す記号である.

この  $\mathcal{U}$  の定義は以下のように  $X$  を入力とする単一化アルゴリズムとして読める:

- $X$  が空集合であれば空の代入を返す.
- そうでなければ,  $X$  から型のペア  $(\tau_1, \tau_2)$  を任意に一つ選び, それ以外の部分を  $X'$  とし,  $(\tau_1, \tau_2)$  がどのような形をしているかによって, 以下の各動作を行う.
  - $\tau_1$  と  $\tau_2$  がすでに同じ形であった場合:  $X'$  について再帰的に単一化を行い, その結果を返せばよい. ( $\tau_1$  と  $\tau_2$  はすでに同じ形なので, 残りの制約集合  $X'$  の解がそのまま全体の解となる.)
  - 選んだ型のペアがどちらも関数型の形をしていた場合, すなわち  $\tau_1$  が  $\tau_{11} \rightarrow \tau_{12}$  の形をしており,  $\tau_2$  が  $\tau_{21} \rightarrow \tau_{22}$  の形をしていた場合:  $\tau_1$  と  $\tau_2$  が同じ形となるためには  $\tau_{11}$  と  $\tau_{21}$  が同じ形であり, かつ  $\tau_{12}$  と  $\tau_{22}$  が同じ形であればよい. これを満たす型代入を求めるために,  $\mathcal{U}$  を  $\{(\tau_{11}, \tau_{21}), (\tau_{12}, \tau_{22})\} \uplus X'$  を入力として再帰的に呼び出し, 帰ってきた結果を全体の結果とする.

<sup>3</sup>問題設定によっては決定不能 (*undecidable*) になることもある. 決定不能であるとは, いい加減に言えば, かつすべての入力について有限時間で停止し正しい出力を返すプログラムが存在しないことを言う. 従って, 決定不能な問題を計算機でなんとかしようとすると, 一部の入力については正しくない答えを返すことを許容するか, 一部の入力については停止しないことを許容しなければならない.

<sup>4</sup>このアルゴリズムは, Prolog などの論理型言語と呼ばれるプログラミング言語の処理系において多く用いられる.

- 選んだ型のペアが型変数と型のペア，すなわち  $(\alpha, \tau)$  か  $(\tau, \alpha)$  の形をしていた場合<sup>5</sup>: この場合，型変数  $\alpha$  は  $\tau$  でなければならないことがわかる．したがって，残りの制約  $X'$  中の  $\alpha$  に  $\tau$  を代入した制約  $[\alpha \mapsto \tau]X'$  を再帰的に解き，得られた解に  $\alpha$  を  $\tau$  に代入する写像  $[\alpha \mapsto \tau]$  を合成して得られる写像  $\mathcal{U}([\alpha \mapsto \tau]X') \circ [\alpha \mapsto \tau]$  を解として返せばよい．ところが，ここで注意すべきことが一つある．もし  $\tau$  中に  $\alpha$  が現れていた場合<sup>6</sup>，ここでエラーを検出しなければならない．(なぜなのかを考察する課題を以下に用意している.)

**Exercise 4.3.3** 上の単一化アルゴリズムを

```
val unify : (ty * ty) list -> subst
```

として実装せよ．

**Exercise 4.3.4** 単一化アルゴリズムにおいて， $\alpha \notin FTV(\tau)$  という条件はなぜ必要か考察せよ．

#### 4.3.4 ML<sup>3</sup> 型推論アルゴリズム

以上を総合すると，ML<sup>3</sup> のための型推論アルゴリズムが得られる．例えば， $e_1 + e_2$  式に対する型推論は，T-PLUS 規則を下から上に読むと，

1.  $\Gamma, e_1$  を入力として型推論を行い， $S_1, \tau_1$  を得る．
2.  $\Gamma, e_2$  を入力として型推論を行い， $S_2, \tau_2$  を得る．
3. 型代入  $S_1, S_2$  を  $\alpha = \tau$  という形の方程式の集まりとみなして， $S_1 \cup S_2 \cup \{(\tau_1, \text{int}), (\tau_2, \text{int})\}$  を単一化し，型代入  $S_3$  を得る．
4.  $S_3$  と  $\text{int}$  を出力として返す．

となる．部分式の型推論で得られた型代入を方程式とみなして，再び単一化を行うのは，ひとつの部分式から  $[\alpha \mapsto \tau_1]$ ，もうひとつからは  $[\alpha \mapsto \tau_2]$  という代入が得られた時に  $\tau_1$  と  $\tau_2$  の整合性が取れているか（単一化できるか）を検査するためである．

**Exercise 4.3.5** 他の型付け規則に関しても同様に型推論の手続きを与えよ (レポートの一部としてまとめよ)．そして，図 4.2 を参考にして，型推論アルゴリズムの実装を完成させよ．

<sup>5</sup> $(\alpha, \alpha)$  の形だった場合はこのケースではなく，一つ前のケースに当てはまる．

<sup>6</sup>繰り返しになるが， $\tau$  が  $\alpha$  自体であった場合はこのケースには当てはまらない．ここでエラーを報告しなければならないのは，例えば  $\tau$  が  $\alpha \rightarrow \alpha$  の場合である．



**Exercise 4.3.6** [★★] 再帰的定義のための `let rec` 式の型付け規則は以下のように与えられる.

$$\frac{\Gamma, f : \tau_1 \rightarrow \tau_2, x : \tau_1 \vdash e_1 : \tau_2 \quad \Gamma, f : \tau_1 \rightarrow \tau_2 \vdash e_2 : \tau}{\Gamma \vdash \text{let rec } f = \text{fun } x \rightarrow e_1 \text{ in } e_2 : \tau} \quad (\text{T-LETREC})$$

型推論アルゴリズムが `let rec` 式を扱えるように拡張せよ.

**Exercise 4.3.7** [★★] 以下は, リスト操作に関する式の型付け規則である. リストには要素の型を  $\tau$  として  $\tau \text{ list}$  という型を与える.

$$\frac{}{\Gamma \vdash [] : \tau \text{ list}} \quad (\text{T-NIL})$$

$$\frac{\Gamma \vdash e_1 : \tau \quad \Gamma \vdash e_2 : \tau \text{ list}}{\Gamma \vdash e_1 :: e_2 : \tau \text{ list}} \quad (\text{T-CONS})$$

$$\frac{\Gamma \vdash e_1 : \tau \text{ list} \quad \Gamma \vdash e_2 : \tau' \quad \Gamma, x : \tau, y : \tau \text{ list} \vdash e_3 : \tau'}{\Gamma \vdash \text{match } e_1 \text{ with } [] \rightarrow e_2 \mid x :: y \rightarrow e_3 : \tau'} \quad (\text{T-MATCH})$$

型推論アルゴリズムがこれらの式を扱えるように拡張せよ.

typing.ml:

```
type subst = (tyvar * ty) list

let rec subst_type subst t = ...

(* eqs_of_subst : subst -> (ty * ty) list
   型代入を型の等式集合に変換 *)
let eqs_of_subst s = ...

(* subst_eqs: subst -> (ty * ty) list -> (ty * ty) list
   型の等式集合に型代入を適用 *)
let subst_eqs s eqs = ...

let rec unify l = ...

let ty_prim op ty1 ty2 = match op with
  Plus -> [(ty1, TyInt); (ty2, TyInt)], TyInt
  | ...

let rec ty_exp tyenv = function
  Var x ->
    (try ([], Environment.lookup x tyenv) with
     Environment.Not_bound -> err ("variable not bound: " ^ x))
  | ILit _ -> ([], TyInt)
  | BLit _ -> ([], TyBool)
  | BinOp (op, exp1, exp2) ->
    let (s1, ty1) = ty_exp tyenv exp1 in
    let (s2, ty2) = ty_exp tyenv exp2 in
    let (eqs3, ty) = ty_prim op ty1 ty2 in
    let eqs = (eqs_of_subst s1) @ (eqs_of_subst s2) @ eqs3 in
    let s3 = unify eqs in (s3, subst_type s3 ty)
  | IfExp (exp1, exp2, exp3) -> ...
  | LetExp (id, exp1, exp2) -> ...
  | FunExp (id, exp) ->
    let domty = TyVar (fresh_tyvar ()) in
    let s, ranty =
      ty_exp (Environment.extend id domty tyenv) exp in
    (s, TyFun (subst_type s domty, ranty))
  | AppExp (exp1, exp2) -> ...
  | _ -> Error.typing ("Not Implemented!")
```

図 4.2: ML<sup>3</sup> 型推論の実装 (2)